Curs pràctic de Maple

Pràctica 11

11 Equacions diferencials

La comanda de Maple que permet solucionar una equació diferencial és `dsolve`. Per a escriure una equació diferencial, es pot utilitzar indistintament la comanda `diff` o la comanda `D`, com es pot veure en els exemples següents:

Exemple 11.1
Si es vol resoldre l'equació diferencial més simple $y' = y$ posarem

```
> dsolve(diff(y(x),x)=y(x),y(x));
```

I si volem afegir-hi una condició inicial bastarà fer

```
> dsolve({diff(y(x),x)=y(x),y(0)=3},y(x));
```

En comptes d'utilitzar `diff` també es pot utilitzar `D` seguint els convenis equivalents.

```
> dsolve({D(y)(x)=y(x),y(0)=3},y(x));
```

Per a practicar, podeu resoldre les equacions diferencials següents:

Exercici 11.1

- $y' - y\tan(x) = 3\exp(-\sin(x))$
- $xy' + y = y^2 x\ln(x)$
- $y' = x^2 - 2xy + y^2$
- $y' = \dfrac{x-y+1}{x+y+3}$
- $y' = \dfrac{x-y-1}{2x-2y+1}$
- $(\exp(y) - 2xy)y' = y^2$
- $(xy' - y)\cos(\frac{2y}{x}) = -3x^4$
- $(2\cos y)y' + \sin y = x^2 \operatorname{cosec}(y)$

Substituint els valors que surten en cada una de les equacions podeu comprovar que les solucions que dóna la comanda `dsolve` són correctes.

I si voleu un exercici amb enunciat (trivial si s'aplica el sentit comú), podeu resoldre el següent:

Exercici 11.2

Els rajos que surten d'una font lluminosa situada a l'origen de coordenades del pla es reflecteixen en una certa corba d'aquest pla i tornen al punt d'emissió. Quina és aquesta corba?

Tots els exemples anteriors són equacions diferencials *de primer ordre* i amb una sola incògnita, com ja deveu suposar la comanda `dsolve` també pot treballar amb equacions d'ordres superiors i amb sistemes d'equacions diferencials amb més d'una incògnita.

Exemple 11.2
El sistema $y' = z, z' = -y$ es pot solucionar amb

```
> dsolve({D(y)(x)=z(x), D(z)(x)=-y(x)},{y(x),z(x)});
```

I si es vol que $y(0) = 0, z(0) = 1$ resultarà

```
> dsolve({D(y)(x)=z(x), D(z)(x)=-y(x),y(0)=0,z(0)=1},{y(x),z(x)});
```

(fet que és ben conegut!).

Per a les equacions d'ordre superior (en les que intervenen derivades d'ordre dos o més gran) es pot treballar també amb `diff` o amb `D`. Veieu a continuació un parell d'exemples (notareu que en la comanda `dsolve` no s'especifica al final la funció respecte de la que volem trobar les solucions, sovint no és necessari).

Exemple 11.3
Primer utilitzant `diff`

```
> dsolve(y(x)*diff(y(x),x$2)-diff(y(x),x)^2-x*y(x)^2=0);
```

I ara utilitzant `D` en un altre exemple

```
> dsolve(y(x)*(D@@2)(y)(x)+(D(y)(x))^2=0);
```

Observareu que per a designar la funció segona derivada de y hem utilitzat l'operador `@@`. Cal dir que, en general, la comanda `f@@n` és una manera ràpida de representar la funció resultant de composar `n` cops la funció `f` amb ella mateixa (en l'exemple `(D@@2)(y)` és sinònim de `D(D(y))`).

Exercici 11.3

Quina és la solució de l'equació diferencial de tercer ordre

$$y''' - 3y'' + y' + y = \exp(-x)(10 - 4x)$$

tal que $y(0) = 5, y'(0) = 6, y''(0) = 3$?

Hi ha moltes equacions diferencials que no tenen solucions en les que la incògnita es pugui expressar de forma simple en funció de la variable independent. L'opció `implicit` de la comanda `dsolve` fa que els procediments de solució de l'equació no intentin aïllar la incògnita en funció de la variable i es conformin amb una expressió implícita.

Exemple 11.4
Proveu per un costat

```
> dsolve((3*y(x)^2+exp(x))*diff(y(x),x)+exp(x)*(y(x)+1)+cos(x)=0,implicit);
```

i per un altre

```
> dsolve((3*y(x)^2+exp(x))*diff(y(x),x)+exp(x)*(y(x)+1)+cos(x)=0);
```

Exercici 11.4

a) Feu un gràfic de la solució de l'equació diferencial de l'exemple anterior que passa pel punt $(3, 5)$.

b) Feu un gràfic en el que es vegin 20 solucions diferents de l'equació de l'exercici anterior.

Una altra forma d'obtenir les solucions d'una equació diferencial és en forma paramètrica (s'obtenen totes les variables com a funcions d'un paràmetre) i d'aquesta forma es pot descriure la *trajectòria* d'una solució. Aquest tipus de solució es pot obtenir combinant les opcions `implicit` i `parametric` en un `dsolve`. Per exemple

Exemple 11.5
Proveu `dsolve` amb l'equació $yy' = -x$ i opcions `parametric` i `implicit`. Feu un gràfic d'algunes de les solucions que s'obtenen. Comproveu que les solucions són les circumferències amb centre a l'origen de coordenades (això serà especialment obvi si deixeu només l'opció `implicit`).

11.1 Gràfics d'equacions diferencials

A part dels gràfics que es poden obtenir amb la comanda `plot` o amb `implicitplot` disposem també de funcions específiques per a fer gràfics a partir d'equacions diferencials dins del paquet **DEtools**. La comanda bàsica és `DEplot` que fa un gràfic del camp de direccions que determina l'equació diferencial com en l'exemple següent:

Exemple 11.6

```
> with(DEtools):
> DEplot(D(y)(x)=4*x/y(x),y(x),x=-2..2,y=-2..2);
```

Podeu veure que s'ha d'escriure l'equació diferencial, la incògnita i la *finestra* en la que s'ha de representar el camp.

Aquesta mateixa comanda també té opcions per a representar solucions concretes de l'equació diferencial (més d'una a l'hora) marcant dins de la comanda la/les condició/ons inicials que ha/n de verificar.

Exemple 11.7
Per a veure una solució marcada sobre el camp es fa

```
> DEplot(D(y)(x)=4*x/y(x),y(x),x=-2..2,[[y(0)=1]],y=-2..2);
```

I si s'en volen veure més es pot fer

```
> DEplot(D(y)(x)=4*x/y(x),y(x),x=-2..2,[[y(0)=1],[y(0)=0.5],[y(0)=0.25]],
> y=-2..2);
```

També es pot fer un dibuix d'un sistema

```
> DEplot(D(x)(t)=y(t),D(y)(t)=-x(t),x(t),y(t),t=0..2*Pi,
> [[x(0)=0,y(0)=1]],x=-2..2,y=-2..2,scaling=constrained);
```

Si només interessa dibuixar algunes de les solucions de l'equació diferencial i eliminar el dibuix del camp, es pot utilitzar l'opció `arrows= none` (l'opció `arrows` també pot tenir assignats el valors `small, medium, large` que canviaran la grandària de les fletxes que representen el camp).

Podeu trobar més opcions d'aquesta comanda amb `?DEplot`.

11.2 Solucions numèriques d'equacions diferencials

Amb la mateixa comanda `dsolve` es poden obtenir solucions numèriques d'una equació diferencial. És suficient afegir l'opció `numèric` com en l'exemple següent:

Exemple 11.8

```
> solnum:= dsolve({D(y)(x)=x^2+y(x)^3,y(0)=0},y(x),numeric);
```

Veureu que en realitat la resposta de Maple no és cap valor numèric concret sinó un *procediment* que calcula aquests valors aproximats de la solució demanada.

```
> seq(solnum(i/10),i=0..10);
```

Es poden fer directament gràfics d'aquestes aproximacions numèriques de solucions d'equacions diferencials amb la comanda `odeplot` del paquet **plots**.

Exemple 11.9

```
> with(plots):
> odeplot(solnum,[x,y(x)],x=-1..1);
```

Finalment, no deixeu de consultar l'ajuda de Maple sobre la comanda `dsolve` (`?dsolve`) i les seves opcions ni la del paquet **DEtools** (`?DEtools`) on podreu veure totes les funcions que posa a la vostra disposició per al tractament de les equacions diferencials.